

인덱스 컬러와 RGB 트루컬러의 상호전환

인덱스 컬러의 활용 분야

인덱스 컬러(Indexed Color)는 최대한 표현할 수 있는 색의 수가 256개로 제한되므로 1670만 가지 색의 트루컬러 만큼 풍부한 색상을 표현할 수는 없지만, 웹에 올리기 위한 사진이나 다이어그램과 같이 많은 컬러가 필요하지 않은 영상을 표출하는 데에는 충분합니다. 게다가 인덱스 컬러 영상의 크기는 트루컬러 영상 크기의 거의 1/3 정도 밖에 되지 않습니다. 이런 이유로 인덱스 컬러 방식을 사용하는 GIF나 PNG 포맷이 웹용 그림 포맷으로 애용되고 있습니다.

인덱스 컬러의 문제

단순화시켜 두 개의 픽셀만을 가지는 영상 배열 [34, 36]을 생각해 보겠습니다. 이를 확대하여 3개의 요소를 가지는 영상으로 만들 때, 단순 내삽을 한다면 일반적으로 [34, 35, 36]을 생각할 수 있습니다. 하지만 이 값이 인덱스 컬러의 색 번호라면 이렇게 내삽을 할 경우 문제가 발생합니다. 34, 35, 36번 색의 연계성이 보장되어 있지 않으며, 그렇기 때문에 35번이 34번과 36번의 중간색이라고 생각할 수 없습니다.

문제 해결 방안

이런 이유로, 인덱스 컬러 공간에서 내삽을 하는 것은 불합리합니다. 인덱스 컬러를 RGB 트루 컬러 공간으로 변환하여 처리하는 것이 좋습니다. 예를 들어, 인덱스 34번과 인덱스 36번의 RGB 색상이 각각 $RGB_{34}=(100, 0, 50)$ 과 $RGB_{36}=(200, 80, 150)$ 이라면, 이를 3개 픽셀로 보간 확대할 때는, 가운데 픽셀을 $RGB=(150, 40, 100)$ 의 색으로 할당하는 것이 옳습니다. 단순화된 예제에서 뿐 아니라, 실제 영상의 Bilinear Interpolation, Cubic Interpolation 등 다양한 보간 처리도 RGB 트루 컬러 공간에서 이루어지는 것이 합당합니다.

인덱스 컬러 → RGB 트루컬러 변환

1. 예제 영상의 다운로드


IDL에서 IDLurlNET 클래스를 이용하여, 기상청 홈페이지에서 제공하는 합성 위성 영상을 다운로드 받는 방법은 다음과 같습니다.(영상의 주소는 아래와 같음)

http://www.kma.go.kr/repository/image/sat/img/MTSAT_LCM_201003290500.gif

```
kmafile='MTSAT_LCM_201003290500.gif'
oURL=obj_new('IDLnetURL', $
    URL_HOSTNAME='www.kma.go.kr', $
    URL_PATH='repository/image/sat/img/'+ $
    kmafile)
```

```
file=oURL->get(filename=kmafile)
obj_destroy, oURL
print, file
;출력 예 D:\MTSAT_LCM_201003290500.gif
```

출력 결과는 로컬디스크에 저장된 경로와 파일명으로, 개인 환경 설정에 따라 저장 디렉토리가 다르게 나옵니다.




기상청(www.kma.go.kr)이 제공하는 위성 영상에 대한 속성 조회

2. 다운로드 영상 디스플레이

인덱스 컬러 영상 포맷인 GIF는 컬러테이블을 적용하지 않으면 원하는 색상이 나오지 않습니다. (흑백 영상으로 본다고 해도 결과에서 글자가 보이지 않을 것입니다).

```
device, decomposed=0
;Indexed Color 모드를 사용하기 위한 선언
read_gif, file, kmaimg
tv, kmaimg
```



컬러테이블을 적용하지 않으면 정상적으로 색이 표출되지 않습니다.

다음과 같이 컬러테이블을 함께 읽어 적용(tvlct)하는 단계가 선행되어야 합니다.

```
read_gif, file, kmaimg, R, G, B
;컬러테이블을 R, G, B 변수로 읽어 옵니다.
tvlct, R, G, B
;컬러테이블을 반영합니다.
tv, kmaimg
```


영상의 크기는 다음과 같이 알아낼 수 있습니다.

```
origsz=size(kmaimg, /DIMENSION)
print, origsz ; 512, 512
```

3. 인덱스 컬러 영상을 바로 확대했을 경우의 문제

```
bigsz=[720, 720] ;720*720으로 확대하고자 할 때
bigimg=congrid(kmaimg, bigsz[0], bigsz[1], $
    /INTERP)
tv, bigimg
```

결과에서 보듯이 인덱스 컬러 영상을 그대로 보간 확대 하게 되면 원하는 결과를 얻을 수 없습니다.



인덱스 컬러 영상을 바로 보간 확대하는 것은 옳지 않습니다. RGB영상을 만들어 확대하면 원하는 결과를 얻을 수 있습니다.

4. 인덱스 컬러 영상 → RGB 트루 컬러 영상으로 변환

인덱스 컬러 영상을 컬러테이블의 적색, 녹색, 청색 값으로 다시 분리하여 RGB 트루컬러 영상의 세 채널에 각각 할당하는 방법으로 간단히 변환할 수 있습니다. 사실상 요즘의 디스플레이는 모두 최종적으로 트루컬러로 표출되므로 인덱스 컬러 영상도 내부적으로는 이러한 변환 과정을 거쳐 출력되고 있다고 할 수 있습니다.

```
rgbimg=bytarr(3, origsz[0], origsz[1])
rgbimg[0,*]=R[kmaimg]
rgbimg[1,*]=G[kmaimg]
rgbimg[2,*]=B[kmaimg]
tv, rgbimg, TRUE=1
```

5. RGB 영상을 확대

RGB 트루컬러 영상은 3개의 채널을 가지는 3차원 배열이므로 확대할 때, (3, Xsize, Ysize) 와 같이 크기를 지정합니다.

```
bigimg=congrid(rgbimg, 3, bigsz[0], bigsz[1], $/INTERP)
tv, bigimg, TRUE=1
```

RGB 공간에서 확대한 영상을 표출해 보면, 보간을 적용했음에도(/INTERP) 원본 색상이 유지되고 있음을 확인할 수 있습니다. 이제, 이 결과를 다시 GIF나 PNG로 저장하기 위해 인덱스 컬러 영상으로 변환하려면 어떻게 해야 할까요?

RGB 트루컬러 → 인덱스 컬러 변환

GIF, PNG 등의 포맷으로 저장하기 위하여, RGB 트루컬러 영상을 인덱스 컬러 영상으로 변환해야 할 때가 있습니다. IDL에서는 COLOR_QUAN() 함수를 이용하여 간단하게 이러한 변환이 가능합니다. Color Quantize를 의미하는 이 함수는 이름 그대로 전체 컬러 공간을 N개의 인덱스 컬러로 요약하고, 이 컬러테이블을 이용하는 1채널 영상(인덱스 컬러 영상)을 생성해 줍니다. 사용이 어렵지 않습니다.

1. RGB 영상 읽어 오기

```
file=filepath('rose.jpg', $
sub=['examples', 'data'])
read_jpeg, file, rose
tv, ROSE, true=1
help, rose
; Array[3, 227, 149]
```



2. COLOR_QUAN() 사용


```
result=color_quan(rose, 1, R, G, B)
; 기본 설정은 256 색 테이블을 만들게 됩니다.
tv!ct, R, G, B
tv, result
```

원본 트루컬러 영상



256컬러로 변환한 영상. 자세히 보면 색의 계조가 불연속적인 곳이 보이긴 합니다. 왼쪽의 청색/보라색 영역을 원본과 비교해 보세요. 하지만 1/3 분량의 데이터로 이만한 표출을 할 수 있다는 것은 큰 의미가 있습니다.

생성한 컬러테이블(R, G, B 변수)을 확인해 보면, 이것이 rose.jpg를 표출하기 위한 컬러테이블이라는 것은 파악할 수 있을 것입니다.



xpalette

컬러 테이블을 보면 이처럼 인접한 색상 간의 연속성이 별로 확인되지 않습니다. 그렇기 때문에 인덱스 컬러 영상을 바로 내삽할 수 없는 것입니다.

rose.jpg 영상을 256색으로 표현하기 위해 생성된 컬러 테이블.

Color_QUAN() 함수의 COLORS 키워드를 이용하여 몇 개의 색을 이용하는 영상으로 변환할 것인지 지정할 수 있습니다. 디폴트는 256 컬러이며 256컬러 이하의 숫자를 지정할 수 있습니다. 예제 영상을 8 컬러 영상으로 만드는 방법은 다음과 같습니다.

```
result=color_quan(rose, 1, R, G, B, COLORS=8)
tv!ct, R, G, B
tv, result
```



8 컬러로 변환한 영상. 자세한 색조는 대부분 소실되었지만 전체적인 구조는 확인이 됩니다.



rose 영상을 8컬러로 표출하기 위한 컬러테이블

3. Indexed 컬러 영상의 저장

대표적인 Indexed Color 영상 포맷인 GIF로 저장하는 방법은 다음과 같습니다. 컬러테이블에 해당하는 값을 함께 저장해 주어야 한다는 점만 눈여겨 보시면 되겠습니다.

```
write_gif, 'rose.gif', result, R, G, B
```